Цепочка Вольтерры

К лекции 9 (2024)

VL: $u_{n,t} = u_n\,(u_{n+1} - u_{n-1})$

Законы сохранения, симметрии.
Несколько простых численных экспериментов.

1 Численный счёт, решение солитонного типа

In[91]:=

```
Clear[u]

(* задаём период и время счёта *)
M1 = 100;
t1 = 1000;

(* уравнения, с учётом периодического замыкания с периодом M *)
eqs = Table[u[n]'[t] == u[n][t] (u[Mod[n + 1, M1]][t] - u[Mod[n - 1, M1]][t]), {n, 0, M1 - 1}];

(* задаём начальные условия. Попробуйте что-нибудь поменять *)
ic = Table[{u[n][0] == 1 + 0.5 Exp[-0.01 (n - M1/2)^2]}, {n, 0, M1 - 1}];

(* строим решение при помощи NDSolve *)
sol1 = NDSolve[Join[eqs, ic], Table[u[n], {n, 0, M1 - 1}], {t, 0, t1}, MaxSteps → ∞][[1]];

(* массив точек *)
u1[t_] := Table[{n, u[n][t]}, {n, 0, M1 - 1}] /. sol1

gr1[t_] := ListPlot[{u1[t], u1[t]},
  PlotRange → {{-0.1, M1 + 1}, {0.95, 2.05}},
  PlotStyle → {{Thin, Black}, {Blue, PointSize[0.01]}},
  Joined → {True, False},
  AxesLabel → {" n", "u_n"},
  BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"},
  ImageSize → 400,
  Epilog → {Text["t = " <> ToString[t], {0.8 M1, 1.95}, {-1, 0}]}
 ]
```

In[99]:=
```
(* анимация *)
Manipulate[gr1[t],
 {{t, 0}, 0, t1, Appearance → {"Labeled", "Open"}, AnimationRate → 5}]
```

Out[99]=



2 Распад пичка

Что получится, если возмутить постоянное решение в одном-единственном узле?

In[108]:=

```
Clear[u]
M2 = 500;
t2 = 1000;

eqs = Table[u[n]'[t] == u[n][t] (u[Mod[n + 1, M2]][t] - u[Mod[n - 1, M2]][t]), {n, 0, M2 - 1}];

ic = Table[{u[n][0] == If[n == Quotient[M2, 2], 1.5, 1]}, {n, 0, M2 - 1}];

(* строим решение при помощи NDSolve *)
sol2 = NDSolve[Join[eqs, ic], Table[u[n], {n, 0, M2 - 1}], {t, 0, t2}, MaxSteps → ∞][[1]];

(* массив точек *)
u2[t_] := Table[{n, u[n][t]}, {n, 0, M2 - 1}] /. sol2

gr2[t_] := ListPlot[{u2[t], u2[t]},
  PlotRange → {{-0.1, M2 + 1}, {0.7, 1.6}},
  PlotStyle → {{Thin, Black}, {Blue, PointSize[0.01]}},
  Joined → {True, False},
  AxesLabel → {" n ", "u_n"},
  BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"},
  ImageSize → 400,
  Epilog → {Text["t =  " <> ToString[t], {0.8 M2, 1.95}, {-1, 0}]}
 ]
```

In[18]:=

```
(* анимация *)
Manipulate[gr2[t],
 {{t, 0}, 0, t1, Appearance → {"Labeled", "Open"}, AnimationRate → 15}]
```

Out[18]=



3 Игра с чётностью

В цепочке Вольтерра есть стационарное решение $u_{2n} = a$, $u_{2n+1} = b$ с разными постоянными a и b. Если немного возмутить его, взяв начальное условие в виде двух плавных кривых, отдельно для чётных и нечётных узлов, то эти узлы так и останутся разделёнными. Конечно, в уравнении КдФ такое невозможно. Для решений такого типа тоже можно написать непрерывный предел, но уже к системе на две полевые переменные.

В этом примере поведение решения кажется более сложным и счёт не такой устойчивый; если сильно увеличить амплитуду возмущений, то в решении могут возникнуть особенности.

In[19]:=

```
M3 = 500;
t3 = 10 000;

eqs = Table[u[n]'[t] == u[n][t] (u[Mod[n + 1, M3]][t] - u[Mod[n - 1, M3]][t]), {n, 0, M3 - 1}];

ic = Table[{u[n][0] ==
     If[OddQ[n],
      1 + 0.1 Exp[-0.005 (n - M3 / 3)^2] + 0.1 Exp[-0.005 (n - 2 M3 / 3)^2],
      (* формула для нечётных узлов *)
      1.2 - 0.05 Exp[-0.005 (n - M3 / 2)^2]]},       (* формула для чётных узлов *)
   {n, 0, M3 - 1}];

sol3 = NDSolve[Join[eqs, ic], Table[u[n], {n, 0, M3 - 1}], {t, 0, t3}, MaxSteps → ∞][[1]];

u3[t_] := Table[{n, u[n][t]}, {n, 0, M3 - 1}] /. sol3
u3o[t_] := Table[{n, u[n][t]}, {n, 1, M3 - 1, 2}] /. sol3
u3e[t_] := Table[{n, u[n][t]}, {n, 0, M3 - 1, 2}] /. sol3

gr3[t_] := ListPlot[{u3[t], u3o[t], u3e[t], u3o[t], u3e[t]},
  PlotRange → {{-0.1, M3 + 1}, {0.85, 1.55}},
  PlotStyle → {{Thin, Black}, {Thin, Black}, {Thin, Black},
    {Blue, PointSize[0.01]}, {Red, PointSize[0.01]}},
  Joined → {True, True, True, False, False},
  AxesLabel → {" n", "u_n"},
  BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"},
  ImageSize → 400,
  Epilog → {Text["t = " <> ToString[t], {0.8 M3, 1.50}, {-1, 0}]}
 ]
```

In[28]:=
```
Manipulate[gr3[t], {{t, 0}, 0, t3, Appearance → {"Labeled", "Open"}, AnimationRate → 10}]
```

Out[28]=



4 Преобразование типа Миуры

Определим оператор линеаризации и разностную вариационную производную.

In[29]:=
```
vars[f_] := Union[Flatten[Cases[f, Blank[#], {0, ∞}] & /@ {u}]]
df[f_] := Plus @@ (D[f, #] × dif[#] & /@ vars[f])
star[f_, g_] := df[f] /. dif[u[k_]] :→ (g /. n → k)

vard[f_] := Expand[Plus @@ (T[D[f, #], n - #[[1]]] & /@ vars[f])]

T[f_, k_] := f /. n → n + k
```

Вариационная производная уничтожает полную разность

In[34]:=
```
vard[u[n] × u[n + 1]]

vard[u[n] - u[n + 1]]

vard[
  (T[#, 1] - #) &[G[u[n - 1], u[n], u[n + 1], u[n + 2]]]
]
```

Out[34]=
```
u[-1 + n] + u[1 + n]
```

Out[35]=
```
0
```

Out[36]=
```
0
```

Определим mVL (пишем u вместо f, U вместо u)

In[37]:=
```
ut = u[n] (u[n] - a) (u[n + 1] - u[n - 1])
```

Out[37]=
```
u[n] (-a + u[n]) (-u[-1 + n] + u[1 + n])
```

По решению f_n этой цепочки определим переменную $u_n = f_n(f_{n+1} - a)$ и проверим, что она удовлетворяет VL

In[38]:=
```
U[n_] := u[n] (u[n + 1] - a)
Factor[star[U[n], ut] - U[n] (U[n + 1] - U[n - 1])]
```

Out[39]=
```
0
```

Точно так же для переменной $u_n = (v_n - a)\, v_{n+1}$

In[40]:=
```
U[n_] := (u[n] - a) u[n + 1]
Factor[star[U[n], ut] - U[n] (U[n + 1] - U[n - 1])]
```

Out[41]=
```
0
```

5 Представление нулевой кривизны

In[42]:=
```
L[n_] := {{0, 1}, {-u[n], λ}}

U[n_] := {{-λ^2/2 + u[n - 1] - 1, λ},
   {-λ u[n], λ^2/2 + u[n] - 1}}

Factor[star[L[n], u[n] (u[n + 1] - u[n - 1])] - U[n + 1].L[n] + L[n].U[n]]
```

Out[44]=
```
{{0, 0}, {0, 0}}
```

6 Плотности законов сохранения

Решаем уравнение (дискретное преобразование Миуры)

$$u_n = f_n\,(f_{n+1} - z)$$

в виде ряда

$$f_n = -\frac{u_n}{z}\left(1 + F_n^{(1)}/z^2 + \ldots\right)$$

Плотности определяются, как коэффициенты разложения

$$\log f_n = -\log(-z) + \log u_n + \log\left(1 + \frac{F_n^{(1)}}{z^2} + \frac{F_n^{(2)}}{z^4} + \frac{F_n^{(3)}}{z^6}\right) = -\log(-z) + \log u_n - r_n^{(1)}/z^2 + r_n^{(2)}/z^4 - r_n^{(3)}/z^6 + \ldots$$

In[45]:=
```
F[0, n_] := 1
F[k_, n_] := -u[n + 1] × Expand[Sum[F[s, n] × F[k - 1 - s, n + 1], {s, 0, k - 1}]] /; k > 0

F[0, n]
F[1, n]
F[2, n]
F[3, n]
```

Out[47]=
```
1
```

Out[48]=
```
-u[1 + n]
```

Out[49]=
```
-u[1 + n] (-u[1 + n] - u[2 + n])
```

Out[50]=
```
-u[1 + n] (u[1 + n]^2 + 2 u[1 + n] × u[2 + n] + u[2 + n]^2 + u[2 + n] × u[3 + n])
```

В принципе, для $r[k]$ можно выписать рекуррентные формулы, но мы просто воспользуемся функцией Series.

In[51]:=
```
M = 6;

FF = Sum[F[s, n] z^(-2 s), {s, 0, M}];

R = Series[Log[FF], {z, ∞, 2 (M - 1)}];

r[0] = Log[u[n]];

Do[r[k] = T[Expand[(-1)^k Coefficient[R, z, -2 k]], -1], {k, 1, 2 (M - 1)}]
```

В результате, получаем такие плотности (если хочется больше, увеличим M):

In[56]:=
```
r[0]
r[1]
r[2]
r[3]
r[4]
r[5]
```

Out[56]=

Log[u[n]]

Out[57]=

u[n]

Out[58]=

$$\frac{u[n]^2}{2} + u[n] \times u[1+n]$$

Out[59]=

$$\frac{u[n]^3}{3} + u[n]^2 u[1+n] + u[n]\, u[1+n]^2 + u[n] \times u[1+n] \times u[2+n]$$

Out[60]=

$$\frac{u[n]^4}{4} + u[n]^3 u[1+n] + \frac{3}{2} u[n]^2 u[1+n]^2 + u[n]\, u[1+n]^3 + u[n]^2 u[1+n] \times u[2+n] + 2 u[n]\, u[1+n]^2 u[2+n] + u[n] \times u[1+n]\, u[2+n]^2 + u[n] \times u[1+n] \times u[2+n] \times u[3+n]$$

Out[61]=

$$\frac{u[n]^5}{5} + u[n]^4 u[1+n] + 2 u[n]^3 u[1+n]^2 + 2 u[n]^2 u[1+n]^3 + u[n]\, u[1+n]^4 + u[n]^3 u[1+n] \times u[2+n] + 3 u[n]^2 u[1+n]^2 u[2+n] + 3 u[n]\, u[1+n]^3 u[2+n] + u[n]^2 u[1+n]\, u[2+n]^2 + 3 u[n]\, u[1+n]^2 u[2+n]^2 + u[n] \times u[1+n]\, u[2+n]^3 + u[n]^2 u[1+n] \times u[2+n] \times u[3+n] + 2 u[n]\, u[1+n]^2 u[2+n] \times u[3+n] + 2 u[n] \times u[1+n]\, u[2+n]^2 u[3+n] + u[n] \times u[1+n] \times u[2+n]\, u[3+n]^2 + u[n] \times u[1+n] \times u[2+n] \times u[3+n] \times u[4+n]$$

Проверяем, что это плотности: дифференцируем в силу VL, должны получиться полные разности. Следовательно, если применить вариационную производную, должны получиться нули.

In[62]:=
```
star[Table[r[k], {k, 0, M - 1}], u[n] (u[n + 1] - u[n - 1])];
vard /@ %
```

Out[63]=

{0, 0, 0, 0, 0, 0}

7 Высшие симметрии

Высшие симметрии определяются как

$$u_{n,t_k} = u_n\left(T - T^{-1}\right)\left(u_n\, \delta\, r_n^{(k)} / \delta\, u_n\right)$$

In[64]:=
```
uT[k_] := Factor[u[n] (T[#, 1] - T[#, -1]) &[u[n] × vard[r[k]]]]

uT[0]
uT[1]
uT[2]
uT[3]
```

Out[65]=

```
0
```

Out[66]=

```
-u[n] (u[-1 + n] - u[1 + n])
```

Out[67]=

```
-u[n]
 (u[-2 + n] × u[-1 + n] + u[-1 + n]^2 + u[-1 + n] × u[n] - u[n] × u[1 + n] - u[1 + n]^2 - u[1 + n] × u[2 + n])
```

Out[68]=

```
-u[n] (u[-3 + n] × u[-2 + n] × u[-1 + n] + u[-2 + n]^2 u[-1 + n] +
    2 u[-2 + n] u[-1 + n]^2 + u[-1 + n]^3 + u[-2 + n] × u[-1 + n] × u[n] + 2 u[-1 + n]^2 u[n] +
    u[-1 + n] u[n]^2 - u[n]^2 u[1 + n] - 2 u[n] u[1 + n]^2 - u[1 + n]^3 - u[n] × u[1 + n] × u[2 + n] -
    2 u[1 + n]^2 u[2 + n] - u[1 + n] u[2 + n]^2 - u[1 + n] × u[2 + n] × u[3 + n])
```

Проверка коммутативности

In[69]:=
```
Expand[star[uT[1], uT[2]] - star[uT[2], uT[1]]]
Expand[star[uT[1], uT[3]] - star[uT[3], uT[1]]]
Expand[star[uT[1], uT[4]] - star[uT[4], uT[1]]]

Expand[star[uT[2], uT[3]] - star[uT[3], uT[2]]]
Expand[star[uT[2], uT[4]] - star[uT[4], uT[2]]]
Expand[star[uT[3], uT[4]] - star[uT[4], uT[3]]]
```

Out[69]=

```
0
```

Out[70]=

```
0
```

Out[71]=

```
0
```

Out[72]=

```
0
```

Out[73]=

```
0
```

Out[74]=

```
0
```

8 Солитон

Проверка связи

In[75]:=
```
Clear[u]
ut = u[n] (u[n + 1] - u[n - 1]);

F = u[n] (u[n + 1] + u[n] + u[n - 1]) + A u[n] + B + (-1)^n C;

Factor[star[F, ut] - u[n] (T[F, 1] - T[F, -1])]
```

Out[78]=
```
0
```

Пара легких упражнений.

1) Определим $u_n(t)$ по явным формулам

$$u_n(t) = \frac{w_{n+1}(t)\, w_{n-2}(t)}{w_n(t)\, w_{n-1}(t)}, \quad w_n(t) = 1 + c\, a^n \exp\!\left(\left(a - a^{-1}\right) x\right)$$

где a и c произвольные постоянные, и проверим, что она удовлетворяет VL и связи.

In[79]:=
```
Clear[a, c, u, w]
w[n_] := 1 + c a^n Exp[(a - a^-1) t];
u[n_] := (w[n + 1] × w[n - 2]) / (w[n] × w[n - 1])
Together[D[u[n], t] - u[n] (u[n + 1] - u[n - 1])]

Together[F /. {A → -B - 3, C → 0}]

Together[% /. B → a + a^-1 + 1]
```

Out[82]=
```
0
```

Out[83]=

$$\frac{(-1 + a)^2\, a^{-2+n}\, (1 + a)\, \left(1 + a + a^2 - a B\right)\, c\, e^{\left(-\frac{1}{a}+a\right) t}}{\left(1 + a^n\, c\, e^{\left(-\frac{1}{a}+a\right) t}\right)\, \left(a + a^n\, c\, e^{\left(-\frac{1}{a}+a\right) t}\right)}$$

Out[84]=
```
0
```

Определим функцию для построения графика решения. Для наглядности, массив выведем два раза: как ломаную и как набор точек. Построим график в манипуляторе, чтобы можно было менять t и параметры.

In[85]:=
```
pl[aa_, cc_, tt_, minu_, maxu_] :=
 ListPlot[
  Evaluate[{#, #} &[Table[{n, u[n] /. {t → tt, a → aa, c → cc}}, {n, -30, 30}]]],
  PlotRange → {minu, maxu},
  Joined → {True, False},
  PlotStyle → {Black, {Blue, PointSize[0.015]}},
  BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"},
  Ticks → {Automatic, Range[0.9, 1.1, 0.05]},
  AxesLabel → {"n", "u"},
  ImageSize → 400]

Manipulate[pl[aa, cc, tt, minu, maxu],
 {{tt, 0, "t"}, -15, 15, Appearance → "Labeled"},
 {{aa, 1.5, "a"}, -5, 5, Appearance → "Labeled"},
 {{cc, 1, "c"}, -3, 3, Appearance → "Labeled"},
 {{minu, .99}, -10, 1},
 {{maxu, 1.11}, 1, 20}
]
```

Out[86]=



2) То же самое для решения

$$u_n = -\frac{b\,(n+d)\,e^{2\,b\,t}}{a+e^{2\,b\,t}} \text{ при чётных } n \text{ и } u_n = \frac{b\,(a\,n+c)}{a+e^{2\,b\,t}} \text{ при нечётных } n.$$

Это решение растет по n, то есть, оно не относится к солитонному типу. Но, тем не менее, это некоторое явное решение.

Out[ ]=

0

In[87]:=

```
Clear[a, b, c, d, v]
u[n_] := If[EvenQ[n], -(b (n + d) Exp[2 b t])/(a + Exp[2 b t]), (b (a n + c))/(a + Exp[2 b t])]

Table[Together[D[u[n], t] - u[n] (u[n + 1] - u[n - 1])], {n, -10, 10}]
```

Out[89]=

{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

In[90]:=

```
Manipulate[
 ListPlot[
  Evaluate[{#, #} &[Table[{n, u[n] /. {t → tt, a → aa, b → bb, c → cc, d → dd}}, {n, -30, 30}]]],
  PlotRange → {-13, 13},
  Joined → {True, False},
  PlotStyle → {Black, {Blue, PointSize[0.015]}},
  BaseStyle → {FontSize → 14, FontFamily → "Times New Roman"},
  Ticks → {Automatic, Range[0.9, 1.1, 0.05]},
  AxesLabel → {"n", "u"},
  ImageSize → 400],
 {{tt, 0}, -5, 5, Appearance → "Labeled"},
 {{aa, 1.5}, 1, 5, Appearance → "Labeled"},
 {{bb, 1.5}, 1, 5, Appearance → "Labeled"},
 {{cc, 1.5}, 1, 5, Appearance → "Labeled"},
 {{dd, 1}, -3, 3, Appearance → "Labeled"}
]
```

Out[90]=